
Scripting is not that hard to learn. The purpose of this tutorial is to


teach you the basics of scripting to let you get started. Once you learn


the basics, you can experiment on your own to learn new things.

















CHAPTER #1 - FUNCTIONS





The most basic part of scripting is something called a function. A function


is nothing more than a set of instructions that tells Starsiege to do


something when something else happens. Here's an example of a basic


function.








function player::onAdd(%player)


{


   


}








This function activates when a player joins the game, hence the name


player::onAdd. The %player is a local variable. A variable could be named


anything, it doesn't really matter. In this case, %player refers to the


player that just joined the game.





What if I wanted something to happen when a player joins the game?


What if I wanted Starsiege to send a message box to the new player to 


greet him? Here's an example of what I could do:








function player::onAdd(%player)


{


   messageBox(%player, "Welcome to the server!");


}








This function would send a message box to the %player saying


"Welcome to the server!" whenever that player joined the server.





Functions are made up of different parts. Basically, it can be divided


up into different pieces:





function





This part is at the beginning of every function. It's purpose is pretty


self-explanatory. It designates that the following text is a function.





player::onAdd(%player)





This part tells what kind of function it is. In this case, it is a function


activates when a new player joins the game. But it could be a number of


things. Here are a few examples of other types of functions & when they 


are activated:





player::onAdd(%player)        when a player joins the game





vehicle::onAdd(%vehicleId)    when a vehicle spawns into the game





vehicle::onAttacked(%attacked, %attacker)    when a vehicle is attacked





vehicle::onDestroyed(%destroyed, %destroyer) when a vehicle is destroyed





vehicle::onScan(%scanned, %scanner)    when a vehicle is scanned





structure::onAttacked(%attacked, %attacker)   when a structure is attacked





structure::onDestroyed(%destroyed, %destroyer)  when a structure is destroyed





structure::onScan(%scanned, %scanner)   when a structure is scanned





player::onRemove(%player)    when a player leaves the server





onMissionStart()    when the mission starts





onMissionLoad()    when the mission loads





onMissionEnd()    when the mission ends








The second part of each function are called the arguments. They give the


variables associated with each individual type of function. Here are a few


examples:





function vehicle::onDestroyed(%destroyed, %destroyer)





function vehicle::onAdd(%vehicleId)





In the first example, the variables used are %destroyed and %destroyer,


because those are the things involved in activating the vehicle::onDestroyed 


function. Notice how the "doer" always goes after the object that was


"done-to" (i.e., the %destroyer always goes after the %destroyed). 


The vehicle::onAdd function, on the other hand, only involves one variable,


the person's vehicle that spawned into the game.





All functions must have an opening curly bracket at the beginning and a 


closing curly bracket at the end. Here's an example of what I mean:





function vehicle::onAdd(%vehicleId)


{


   


}





Between these brackets is where we put our commands, the things that we want


to happen. In the following example, whenever a vehicle spawns into the game,


the player's vehicle is destroyed (by the use of the healObject(); command).


The healObject(); command is normally used to heal vehicles. But it can also


be used to destroy a vehicle, as demonstrated below:





function vehicle::onAdd(%vehicleId)


{


   healObject(%vehicleId, -50000);


}





The healObject(); command does the action to the %vehicleId (the player's 


vehicle). The number follows with a comma between to separate it. The number


refers to the amount of healing that is done. Notice the negative number in


the healObject(); command. The negative number is what allows us to destroy


the vehicle instead of heal it. If it were a positive number instead, it


would heal the vehicle that amount instead of hurting it by that amount.


You can experiment with different numbers for different amounts of damage


or healing. Note: A positive sign "+" is not needed before a number value.  





That is only one example of a command. Commands are all written basically


the same way. They all have a command name, a pair of parentheses, and ALL


COMMANDS END IN A SEMI-COLON. If you forget the semi-colon, you will get


a syntax error in your script at the line that the error occurs in. Commands


are placed inside a function. You can have more than one command in a single


function. In fact, you can have several hundreds if you really wanted to.





This is a basic representation of how you put a function together:





function functiontype(%variable)


{


   Place commands here


}








The following example is a function that activates whenever a vehicle


spawns into the game. First, it destroys the player's vehicle, then it


says a message out loud to Everybody in the chat box.





function vehicle::onAdd(%vehicleId)


{


   healObject(%vehicleId, -50000);


   say("Everybody", 1, "Someone tried to spawn unsuccessfully, hehe");


}








Some functions can be used for more specific purposes. What if we wanted


to use a structure::onScan function, but we didn't want it to apply to all


structures in the map? We might want the function to only activate whenever


a certain structure is scanned. For example, if we set the structure's


script class to "pinball" in the mission editor, we can then write a


function that only works when the "pinball" structure is scanned.


(Note: the structure's script class can be named anything you want it to be, 


"pinball" is just an example) Here's an example of what the function might


look like:





function pinball::structure::onScan(%scanned, %scanner)


{


   healObject(%scanner, 10000);


}








The new function was created by merely adding the script class name of the 


structure at the front of the function name, followed by 2 colons. This


method of making specific functions can also be applied to AI vehicles as


well as structures:





function mike::vehicle::onDestroyed(%destroyed, %destroyer)


{


   say("Everybody", 1, "Mike the AI unit has been destroyed!!!");


}








There's also another type of function, called a trigger function.


These functions let you assign actions to triggers that you place in your


map. For example, if I made a trigger in the mission editor and named the


script class as "monkey", I could assign different actions to this new


trigger by using the same principle we just learned. What if I wanted to


make something happen when a vehicle enters the trigger? I would use


something like this:





function monkey::trigger::onEnter(%trigger, %vehicleId)


{


   setPosition(%vehicleId, 250, -400, 200);


}





This function teleports a vehicle to certain coordinates whenever it enters


the trigger.





What if we wanted a function that continually runs as long as the vehicle


remains inside the trigger? We would use a trigger::onContact function,


which is run for as long as the vehicle remains IN CONTACT with the


trigger:





function monkey::trigger::onContact(%trigger, %vehicleId)


{


   healObject(%vehicleId, 2500);


}





This function heals a vehicle for as long as he remains inside the trigger.


If we go into the Mission Editor and set the trigger's contact rate to 1, 


that means that the function is activated once for every second that the


vehicle remains inside the trigger. So the vehicle is healed 2500 for every


second it remains inside the trigger. If we change the trigger's contact


rate to 0.25, the function is activated every 0.25 seconds, thus the vehicle


gets healed 4 times each second.





We've already talked about the trigger::onEnter function, but what if we


want a function that will activate when a vehicle leaves a trigger?


We can use a trigger::onLeave function.





function monkey::trigger::onLeave(%trigger, %vehicleId)


{


   say("Everybody", 1, "Have a nice day!");


}





This function says "Have a nice day!" when a vehicle leaves the "monkey"


trigger.








On a side note, I think I should warn you about placing new triggers into


the mission editor. Whenever you place a new trigger into your map, go to


its properties menu (F2) and make sure the "Is Sphere" box is unchecked.


If your trigger doesn't have it unchecked, it may cause problems with your


trigger, especially if you want to make it bigger than a normal heal pad


trigger (40 40 40).











Functions can do whatever you want them to. As I said before, they are


nothing but sets of instructions for Starsiege that are activated at


particular times.

















CHAPTER #2 - VEHICLE AND PLAYER ID's





There are many types of variables. For now, we will concentrate on


player ID's and vehicle ID's. A player ID represents a particular player.


A vehicle ID represent's the player's vehicle. Some functions use playerID's


and others use vehicleID's. This makes sense, since a function such as


player::onAdd(%player) is activated when a player joins the game. Likewise, 


a function such as vehicle::onAdd(%vehicleId) is activated when a person's


vehicle is spawned into the game.





Sometimes this poses a problem. For example, the healObject() command only


works on vehicle ID's, because obviously you cannot heal a player....only


his vehicle can get healed. Likewise, the messageBox() command only works


for player ID's, because obviously you cannot send a message to a 


vehicle....only to the player. It only makes sense that way. 





Most functions only contain variables with either a player ID or a


vehicle ID, but not both. So we need a way to convert a vehicle ID into


its corresponding player ID and a player ID into its corresponding 


vehicle ID. Fortunately, there are a couple of commands that can do that:





%vehicleId = playerManager::playerNumToVehicleId(%player);





and...





%player = playerManager::vehicleIdToPlayerNum(%vehicleId);





The first command converts a player ID (%player) into a


vehicle ID (%vehicleId). This command is rarely used. The second command,


however, is used quite often. It converts a vehicle ID (%vehicleId) into


a player ID (%player). Take note of this command, because it may help you


later on in the future. Here's an example of how the second command can be


used:





function missionBoundary::trigger::onEnter(%trigger, %vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   messageBox(%player, "WARNING: You have left the mission area!");


}





Notice how the %vehicleId at the beginning of the function (which represents


the person's vehicle) is converted into the %player (which represents


the actual player himself). After we have converted it to %player, we can


use it to send a message box to the player as shown above.





Note: Once you have converted a vehicle ID (%vehicleId) to a player ID


(%player), you can still use the original %vehicleId in your function.


In fact, you can use both the %vehicleId and the player ID (%player).


Here's an example:





function missionBoundary::trigger::onEnter(%trigger, %vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   messageBox(%player, "WARNING: You have left the mission area!");


   healObject(%vehicleId, -50000);


}





The function above is activated when a vehicle enters the "missionBoundary"


trigger. So we converted the %vehicleId into %player, and then sent a 


message box to the %player saying that he'd left the mission area. After 


that, we destroyed his vehicle using the healObject() command.








Some examples of vehicle ID's are as follows:





%vehicleId


%attacked


%attacker


%destroyed


%destroyer


%scanned


%scanner


%object





Each of them is used in a different type of function, yet they still are


all examples of vehicle ID's. Take %attacker for example, it is used in 


an onAttacked function...and the only thing that the attacker can be is a


vehicle. Therefore %attacker is a vehicle ID.





You must use the correct form of the vehicle ID in a command depending on


type of function you are using. For example, although we have only used the 


healObject() command with %vehicleId, it can be used with any form of a


vehicle ID. Here's an example:








function vehicle::onAttacked(%attacked, %attacker)


{


   healObject(%attacked, -50000);


}








This function destroys a vehicle whenever it is attacked. Basically, this


is a function for one-hit kills. Notice that we use %attacked instead of


the usual %vehicleId. We do this because %attacked is one of the 2 variables


involved in the function, therefore we cannot use %vehicleId. We must use


either %attacked or %attacker, depending on which vehicle we want to destroy.


This principle applies to all commands. Always use ONLY the variables that


are involved in that type of function. Here's an example of WHAT NOT TO DO:








function vehicle::onScan(%scanned, %scanner)


{


   healObject(%vehicleId, -50000);                  //this is wrong!


}








The command inside this function uses %vehicleId, which is wrong, because


the variable refers to nothing. An error like this would cause the command


in this function to not work at all. The command can be fixed by changing


%vehicleId to either %scanned or %scanner, because these are the only


2 variables involved in the function.











The only player ID that I can think of is %player. Usually player ID's


are only involved in player::onAdd (when a player joins the server) and


player::onRemove (when a player leaves the server).





Note: If a variable in a function refers to a structure, it is a vehicle ID.


Even though the structure is not a vehicle, it has a vehicle ID. The same


principle applies to turrets and AI units as well. Basically, any object


that can be put in the mission editor has a vehicle ID.

















CHAPTER #3 - BASIC LIST OF COMMANDS











The say(); command:





say("Everybody", 1, "Insert Message Here", "soundfile.wav");





"Everybody" - This part specifies who gets the message. This can also be


replaced with %player.





1 - This part is the channel that the text uses for using the chat window.


It is best to keep this value at 1.





"Insert Message Here" - This part is the message that is sent.





"soundfile.wav" - This part is the sound file (.wav) that you want to play


along with the message. Note: A sound file is not needed to send a message.


Here's an example of the say() command without sound:





say("Everybody", 1, "Insert Message Here");





Just leave the sound file part off the end as shown above.











The healObject(); command:





healObject(%vehicleId, 10000);





%vehicleId - This is the vehicle ID of the vehicle that you want to heal.





10000 - This part is the amount that you want to heal the vehicle. If this


value is a negative number, it hurts instead of heals.











The setPosition(); command:





setPosition(%vehicleId, 150, -200, 100);





%vehicleId - This is the vehicle or structure that you want to teleport to 


a new location.





150 - This part is the x coordinate of the desired destination. It can be


any value you want.





-200 - This part is the y coordinate of the desired destination. It can be


any value you want.





100 - This part is the z coordinate of the desired destination. It can be


any value you want.











The messageBox(); command:





messageBox(%player, "Insert Message Here");





%player - This is the player ID of the player that you want to send the 


message box to. If this is replaced with 0, then the message box is sent


to all players in the server.





"Insert Message Here" - This part is the message that you want to send to


that player.











The reloadObject(); command:





reloadObject(%vehicleId, 1000);





%vehicleId - This is the vehicle ID of the vehicle that you want to reload.





1000 - This is the amount of ammo you want to reload them with.


Note: Larger shells such as blast cannon ammo require a higher number to


reload the same amount. (5 regular ammo = 1 blast cannon ammo)











The blast(); command:





blast(%vehicleId, 150, 2500);





%vehicleId - This is the vehicle or structure that you want to be at the 


center of the blast radius.





150 - This part is the size of the blast radius. The value is in meters. 


It can be changed to any number you want it to be.





2500 - This part is the amount of damage that the blast causes. This can


also be changed to any number you want it to be.











The setTeam(); command:





setTeam(%vehicleId, *IDSTR_TEAM_BLUE);





%vehicleId - This is the vehicle or structure whose team you want to change.





*IDSTR_TEAM_BLUE - This is the desired team that the %vehicleId will be set


to. Other examples of teams include: *IDSTR_TEAM_YELLOW, *IDSTR_TEAM_BLUE, 


*IDSTR_TEAM_RED, *IDSTR_TEAM_PURPLE, *IDSTR_TEAM_NEUTRAL

















CHAPTER #4 - CHECKING FOR CERTAIN CONDITIONS





Now we will learn how to check for certain situations in a function.


There is something called an if() check. It is used inside of a function


to check if a certain condition is met, and if it is, it runs the commands


inside the if() statement. Here's an example of where it is placed inside


the function:








function vehicle::onAdd(%vehicleId)


{


   if()


   {


      (Put a command here)


   }


}








Inside the if() part, we put the conditions we want to check for. What if


we want to check to see if the %vehicleId's HUD name is


"Com. Sentinal [M.I.B.]"?








function vehicle::onAdd(%vehicleId)


{


   if(getHudName(%vehicleId) == "Com. Sentinal [M.I.B.]")


   {


      (Put a command here)


   }


}








The == is used to check to see if 2 things are equal. In this case, we are


checking to see if the HUD name of the %vehicleId equals 


"Com. Sentinal [M.I.B.]". On the contrary, if we wanted to check if 2 things


are NOT equal, we use != instead:








function vehicle::onAdd(%vehicleId)


{


   if(getHudName(%vehicleId) != "Com. Sentinal [M.I.B.]")


   {


      (Put a command here)


   }


}





In this example, we check to see if the HUD name of the %vehicleId DOESN'T


equal "Com. Sentinal [M.I.B.]". If the condition is met, the command inside


the if() statement is run. So in this case, if the person's name isn't


Com. Sentinal [M.I.B.], then the command inside the if() statement will


run.





Here's another example. What if we have a function:








function vehicle::onAdd(%vehicleId)


{


   setTeam(%vehicleId, *IDSTR_TEAM_PURPLE);


}








This function will change a person's team color when their vehicle spawns


into the game. But what if we want to only change the team of people who


join red team? We could do an if() check to find out whether or not they


are on red team:








function vehicle::onAdd(%vehicleId)


{


   if(getTeam(%vehicleId) == *IDSTR_TEAM_RED)


   {


      setTeam(%vehicleId, *IDSTR_TEAM_PURPLE);


   }


}








This function first checks to see if the person is on red team. If they


are, it changes their team to purple. If they are not, then the command 


to change teams does not affect them. What if we wanted it so that


anyone NOT on blue team would automatically have their team set to blue


when they spawned? Here's what we could do:








function vehicle::onAdd(%vehicleId)


{


   if(getTeam(%vehicleId) != *IDSTR_TEAM_BLUE)


   {


      setTeam(%vehicleId, *IDSTR_TEAM_BLUE);


   }


}





While == means "equal to", != means "not equal to".


Here are a few others that can be used (especially with number values):





> "greater than"


< "less than"


>= "greater than or equal to"


<= "less than or equal to"











Now what if we wanted to check for more than one condition to be met?


We could use extra parentheses and some double && signs to do this:








function kill::trigger::onEnter(%trigger, %vehicleId)


{


   if((getHudName(%vehicleId) == "Killer") && (getSquad(%vehicleId) == "NTDF"))


   {


      healObject(%vehicleId, -50000);


   }


}








This function is a trigger::onEnter function that checks for 2 conditions


to be met: the HUD name of the %vehicleId must be "Killer" and his squad


must be "NTDF". If these conditions are met, then that unlucky person's


vehicle gets destroyed. 





!= can also be used instead of ==, so that everybody who walks into the


trigger EXCEPT NTDF Killer gets destroyed:








function kill::trigger::onEnter(%trigger, %vehicleId)


{


   if((getHudName(%vehicleId) != "Killer") && (getSquad(%vehicleId) != "NTDF"))


   {


      healObject(%vehicleId, -50000);


   }


}








Now lets review a few of the commands we can use to check things with:








getHudName(%vehicleId)  - This gets the HUD name of the %vehicleId.





getTeam(%vehicleId)  - This gets the team color of the %vehicleId.





getSquad(%vehicleId)  - This gets the squad name of the %vehicleId.





Here are few more:





getName(%player)  - This gets the name of the %player.





getVehicleName(%vehicleId)  - This gets the type of vehicle that the person


is currently using.





getPosition(%object, x)  - This gets the X coordinate of the object. The


object can be either a vehicle or a structure.





getPosition(%object, y)  - This gets the Y coordinate of the object. The


object can be either a vehicle or a structure.





getPosition(%object, z)  - This gets the Z coordinate of the object. The


object can be either a vehicle or a structure.





getDistance(%object1, %object2)  - This gets the distance (in meters)


between 2 different objects. These objects can be anything from vehicles


to structures to nav points. Anything except %player really. Here's a


quick example on how to use getDistance() correctly:





function structure::onScan(%scanned, %scanner)


{


   if(getDistance(%scanned, %scanner) < 100)


   {


      say("Everybody", 1, "You are close enough to begin download sequence.");


   }


}  





This function is a structure::onScan function that checks to see if your


vehicle is within 100 meters of the structure you scanned. If it is, it


runs the command inside the if() statement. In this case, it sends you


a message saying "You are close enough to begin download sequence", but


you can use any of the commands you have learned.








Now that you've learned the basics of checking for certain conditions,


we can expand on this by learning how to check for several possibilties.


We can use an if() to check for one possibility, but what if we want to


check for more than one possibility? We can use else if() to check for


different possibilites other than the one in the initial if(). 


Consider the following function: 








function vehicle::onAdd(%vehicleId)


{


   if( Possibility #1 )


   {


      Command #1


   }


   else if( Possibility #2 )


   {


      Command #2


   }


   else if( Possibility #3 )


   {


      Command #3


   }


}








This is the basic layout of how to check for more than one possibility.


You can check for as many possibilities as you want. Here's an example


of how this can be used:








function vehicle::onAdd(%vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);





   if(getVehicleName(%vehicleId) == "Goad")


   {


      messageBox(%player, "Goads are okay to use in this server.");


   }


   else if(getVehicleName(%vehicleId) == "Executioner")


   {


      messageBox(%player, "Executioners are not allowed in this server.");


      healObject(%vehicleId, -50000);


   }


   else if(getVehicleName(%vehicleId) == "Prometheus")


   {


      messageBox(%player, "HACKS WILL NOT BE TOLERATED!");


      kick(%player);


   }


}








In this vehicle::onAdd function, we check to see what kind of vehicle the


person has spawned in. If it is a goad, nothing happens to the person and


he just gets a friendly message box. If it is an executioner, we send a


message box to the person telling him that executioners are not allowed 


and we destroy his vehicle. However if the person is using Prometheus,


we send the player a message box saying "HACKS WILL NOT BE TOLERATED",


and we kick him from the server.





There is also an else check. This check is similar to else if() but is


used in a slightly different way. The else check is used whenever


none of the other conditions are met.








function vehicle::onAdd(%vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);





   if(getVehicleName(%vehicleId) == "Prometheus")


   {


      messageBox(%player, "Prometheus is not allowed in this server!");


      healObject(%vehicleId, -50000);


   }


   else if(getVehicleName(%vehicleId) == "Pouncer Assault Bike")


   {


      messageBox(%player, "Bikes are not allowed in this server!");


      healObject(%vehicleId, -50000);


   }


   else if(getVehicleName(%vehicleId) == "Starsiege Magic Bus")


   {


      messageBox(%player, "Buses are not allowed in this server!");


      healObject(%vehicleId, -50000);


   }


   else


   {


      messageBox(%player, "Thank you for not using a hack.");


   }


}





In this vehicle::onAdd function, we checked to see what kind of vehicle


the person spawned into the game with. If the vehicle is one of those


3 hacks checked for above, the person's vehicle gets destroyed. But if


the vehicle is anything else, then nothing happens to the person's vehicle


and the player gets a friendly message box saying "Thank you for not


using a hack."





If() checks can also be used inside each other as well, as shown below:





function vehicle::onAdd(%vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);





   if(getVehicleName(%vehicleId) == "Prometheus")


   {


      if(getHudName(%vehicleId) == "Mike")


      {


         say(%player, 1, "You have clearance to use this hack, Mike.");


      }


      else


      {


         messageBox(%player, "Only Mike can use Prometheus!");


         healObject(%vehicleId, -50000);


      }


   }


}





This vehicle::onAdd function checks to see if the person is using 


Prometheus, then it checks to see if the person's name is "Mike". If it


is, then that person is allowed to use Prometheus. If the person's name


is not Mike, then the person is sent a warning message box and killed


for trying to use Prometheus. Basically, this function only allows a


person named "Mike" to use Prometheus....all other people who try to


use Prometheus are killed.

















CHAPTER #5 - GLOBAL AND LOCAL VARIABLES





What are variables? Variables can take on any value you want, whether it


be a number, a line of text, or a word. There are 2 main types of variables


used in Starsiege scripting: Global variables and Local variables.





Local variables are variables that begin with a percent sign and can only 


be used in one particular function. Local variables cannot be used 


throughout the script, they can only be used in the function they 


originated in (or were defined in). Up until now, all of the variables we


have dealt with have been local variables. The following function contains


several examples of local variables:








function teleport::trigger::onEnter(%trigger, %vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   %x = 150;


   %y = 200;


   %z = 100;


   setPosition(%vehicleId, %x, %y, %z);


   messageBox(%player, "You have been teleported to new coordinates.");


}








All of the variables in the function above are local variables, which means


that they can only be used in that function. The values of these variables


do not matter in another function. Local variables can be classified into


3 general categories: player ID's, vehicle ID's, and constant variables.


We have already talked about player ID's and vehicle ID's. In the function


above, %player is the only player ID. The vehicle ID's in the function


include %vehicleId and %trigger (%trigger is considered a vehicle ID


because it represents a physical object: the trigger itself). The constant


variables in the function include %x, %y, and %z.





Unlike player ID's and vehicle ID's, constant variables can be given a


value. Constant variables can be given any value you want to give them.


Let's take another look at the last function:








function teleport::trigger::onEnter(%trigger, %vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   %x = 150;


   %y = 200;


   %z = 100;


   setPosition(%vehicleId, %x, %y, %z);


   messageBox(%player, "You have been teleported to new coordinates.");


}








Notice how we use the equals sign (=) to give the variables their values.


Then later on down the function, we use our variables in the place of


the numbers normally used in the setPosition() command. These numbers are


the values of the x, y, and z coordinates of the desired destination.


We gave these variables the names %x, %y, and %z because it describes what


they stand for. It doesn't really matter what we name our variables though.


We could just have easily done it differently:








function teleport::trigger::onEnter(%trigger, %vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   %variableName1 = 150;


   %variableName2 = 200;


   %variableName3 = 100;


   setPosition(%vehicleId, %variableName1, %variableName2, %variableName3);


   messageBox(%player, "You have been teleported to new coordinates.");


}








We could have given these variables any name we wanted and they would still


work. You could name your variable %lightBulb for all I care, it doesn't


matter!





Now we've already seen that we can give variables numerical values, but


we can also give them text values as well:








function structure::onScan(%scanned, %scanner)


{


   %switch = "off";


   if(%switch == "off")


   {


      %switch = "on";


      say("Everybody", 1, "Power-up initiated.");


   }


}








Now this function is pretty pointless. I was just trying to make a point


that variables can be given text values too. Notice that the text must have


quotation marks around it for it to work properly.





Constant variables can also be given command values as well. Consider 


the following function:








function randomTeleport::trigger::onEnter(%trigger, %vehicleId)


{


   %destination = randomInt(1,2);


   if(%destination == 1)


   {


      setPosition(%vehicleId, 100, 200, 300);


   }


   else if(%destination == 2)


   {


      setPosition(%vehicleId, 850, 625, 175);


   }


}








In this trigger::onEnter function, we create a variable called %destination.


We give this variable a random value by using the randomInt(); command. 


In this case, the randomInt() command can choose random integers between


1 and 2. So basically, %destination is randomly given a value of either


1 or 2 every time that this function is run. Then we use an if() to check


if %destination equals 1. If it does, then the vehicle is transported to


the coordinates 100, 200, 300. If %destination equals 2 instead, the vehicle


gets transported to different coordinates: 850, 625, 175. So basically, this


function is for a trigger that randomly teleports the person's vehicle to


one of 2 possible destinations.





What if you wanted to use the value of a variable in a message?


Consider the following example:








function ruler::structure::onScan(%scanned, %scanner)


{


   %player = playerManager::vehicleIdToPlayerNum(%scanner);


   %distance = getDistance(%scanned, %scanner);


   say(%player, 1, "You are " @ %distance @ " meters away.");


}








In this example, we set %distance equal to the distance between the


%scanned and the %scanner. Then we send a say() message to the %player,


telling him how far away from the structure he is. We used the @ signs


to connect the value of the variable %distance to the rest of the message,


which is surrounded by quotations. No matter what the value of %distance


turns out to be, it will say that number in the middle of the message.


Think of this as a blank in the middle of a sentence. We just filled


it in with the variable %distance. Using this method, we are able to


send the player a message that contains an indefinite/non-constant value.


Because the distance between the structure and the %scanner is able to


change, we used %distance inside the message to represent that value.


So, for example, if the vehicle was 258 meters away from the structure


when he scanned it, the message he would recieve would say "You are


258 meters away." Also, notice the spaces before the quotations. Their


purpose is to space the value from the words. If you didn't space it,


you may end up with a message looking like this: "You are258meters away".


So always be sure to space.








Global variables are the type of variables that begin with a dollar sign ($).


Unlike local variables, global variables can be used throughout the script.


They can can used in any function in the entire script. Global variables


always have a definite value. Whenever a value of a global variable


changes, it changes in all functions that use that variable. Here's an


example of how a global variable might be used in a script:








function onMissionStart()


{


   $padReady = true;


}





function InstantHealPad::trigger::onEnter(%trigger, %vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);





   if($padReady == true)


   {


      $padReady = false;


      healObject(%vehicleId, 50000);


      healObject(%vehicleId, 50000);


      healObject(%vehicleId, 50000);


      healObject(%vehicleId, 50000);


      healObject(%vehicleId, 50000);


      schedule("$padReady = true", 30);


   }


   else if($padReady == false)


   {


      say(%player, 1, "The pad is not ready yet.");


   }


}








In this short script, there are 2 functions. In the onMissionStart()


function, we set the global variable $padReady equal to true, because


the healpad is ready when the mission starts. Then we have the


InstantHealPad::trigger::onEnter function. In this function, we check


to see if $padReady equals true or false. If it equals true, then $padReady


is set to false, the person's vehicle is healed, and we schedule $padReady


to be set back to true in 30 seconds. If the person's vehicle enters the


heal trigger when $padReady still equals false, the player recieves a


message telling him that the pad isn't ready yet. After the 30 seconds


are up, $padReady is set back to true again and the pad is ready to be


used once again. This is just one example of how global variables can


be useful in scripting.





Global variables are usually given a value of true or false, but they


can take on text and numerical values as well. In this case, the


global variable takes on a text value:





$globalVariable = "text";





Notice how the text value has quotations around it. If the global variable


equals true or false, then quotation marks are not needed. But for all


other text, quotation marks are necessary.





Global variables can also take on a numerical value:





$globalVariable = 7;





In this case, quotation marks are not needed because the variable takes on


a numerical value.








Variables can also be set equal to a vehicle, player, or structure.


In the following script, we set a global variable equal to a person's 


vehicle:








function onMissionStart()


{


   $it = "Nobody";


}





function vehicle::onAdd(%vehicleId)


{


   if($it == "Nobody")


   {


      $it = %vehicleId;


      say("Everybody", 1, getHudName(%vehicleId) @ " is now IT!");


   }


}





function vehicle::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $it)


   {


      if(%destroyed != %destroyer)


      {


         $it = %destroyer;


         say("Everybody", 1, getHudName(%destroyer) @ " is now IT!");


      }


      else if(%destroyed == %destroyer)


      {


         $it = "Nobody";


         say("Everybody", 1, "IT has committed suicide!");


      }


   }


}








This short script is basically a game of tag. When the mission starts,


nobody is IT. Whenever someone spawns and nobody is IT, that person


becomes IT. If somebody kills this person, then they become IT.


And if the person who is IT kills himself, then nobody is IT again.


Because nobody is IT, then whoever respawns first becomes IT. This game


was very easy to script. The script only consisted of 3 simple functions,


yet it was an entirely new game. Soon, when you get the hang of scripting,


you will be able to make your own games. And global & local variables can


be very useful in any script.








On a side note, you can use +, -, *, and / to add, subtract, multiply, and


divide numerical values, whether they be variables with numerical values


or actual numbers. For example if %number1 = 20, %number2 = 3,


%number3 = 10, and:





%answer = (%number3 + %number1) / %number2;





then:





%answer = (10 + 20) / 3;





which means:





%answer = 10;








This is just one example of how to use mathematical symbols in your


scripting. For your reference:





+ means add.


- means subtract.


* means multiply.


/ means divide.

















CHAPTER #6 - ADVANCED COMMANDS





Here is a short list of some advanced commands. Be sure you know how to


use these commands because they will be used in the tutorial from this


point on.











The getTerrainHeight(); command:





getTerrainHeight(%x, %y);





%x - This represents the desired x coordinate value. It can be any number.





%y - This represents the desired y coordinate value. It can be any number.








Example:





getTerrainHeight(100, -370);








This command gets the height of the terrain at the specified x and y 


coordinates. The height is actually the z coordinate of the terrain at


the specified x and y coordinates. This command is usually used in an if()


check or can be the value of a particular variable. Here's 2 examples of 


how it can be used:








if(getTerrainHeight(-20, 450) < 150)





This may not be a very useful way to use this command, but it can be done.








%terrainHeight = getTerrainHeight(175, 120);





This is the most useful way to use the getTerrainHeight() command.


Setting it equal to a variable can be very useful. Here's an example:








function randomTeleport::trigger::onEnter(%trigger, %vehicleId)


{


   %x = randomInt(-500, 500);


   %y = randomInt(-500, 500);


   %z = getTerrainHeight(%x, %y);


   setPosition(%vehicleId, %x, %y, %z);


}








In this example, we found random x and y values between -500 and 500.


Then no matter what the values of %x and %y turned out to be, we find


the terrain height at those 2 coordinates. That value is then set equal


to %z. Then we setPosition() the %vehicleId to those 3 coordinates.


So basically, this function randomly transports a vehicle to some random


coordinates between -500 and 500 in the x and y axes. No matter what those


2 values are, %z will be the z coordinate of the height of the terrain at


those random coordinates. So no matter where the vehicle teleports to,


he will be above the terrain.











The dropPod(); command:





dropPod(%x1, %y1, %z1, %x2, %y2, %z2, %vehicleId);





%x1 - This represents the starting X coordinate of the drop pod. This value


can be any number.





%y1 - This represents the starting Y coordinate of the drop pod. This value


can be any number.





%z1 - This represents the starting Z coordinate of the drop pod. This value


can be any number.





%x2 - This represents the ending X coordinate of the drop pod. This value


can be any number.





%y2 - This represents the ending Y coordinate of the drop pod. This value


can be any number.





%z2 - This represents the ending Z coordinate of the drop pod. This value


can be any number, usually this coordinate is at the terrain height.





%vehicleId - This is the vehicle that you want to drop using the dropPod()


command. This is not required, you can drop a drop pod without dropping


an actual vehicle. If you want just drop the pod itself (without a vehicle),


then merely omit the %vehicleId part of the command. When this command is


used with a vehicle, it is usually only used with AI's. I highly recommand


that you only use AI's with this command. Using a player's vehicle has a


tendency to crash the game. We will talk about using this command with AI's


later in the tutorial.


 





Examples:





dropPod(120, 200, 500, 150, 250, 100, %vehicleId);





OR





dropPod(120, 200, 500, 150, 250, 100);








Basically, the dropPod() command drops a drop pod with or without a vehicle.


Here' an example of how this command can be put to use:








function structure::onScan(%scanned, %scanner)


{


   %x1 = getPosition(%scanner, x) + 100;


   %y1 = getPosition(%scanner, y) + 100;


   %z1 = getTerrainHeight(%x1, %y1) + 500;


   %x2 = getPosition(%scanner, x) + 100;


   %y2 = getPosition(%scanner, y) + 100;


   %z2 = getTerrainHeight(%x2, %y2);


   dropPod(%x1, %y1, %z1, %x2, %y2, %z2);   


}








In this structure::onScan function, we make an empty dropPod drop nearby


the %scanner. We get the coordinates of the %scanner, then offset the


numbers by 100 meters, which is close enough to be seen easily yet not


fall directly on top of the %scanner. Notice how we used getTerrainHeight()


to find the height of the terrain at those particular coordinates.





Basically, the first 3 x, y, and z coordinates is where the drop pod starts


and the last 3 coordinates is where it impacts. If there is a vehicle inside


the pod, then it would be teleported to the impacted area.











The cloneVehicle(); command:





cloneVehicle(%vehicleId);








%vehicleId - This is the vehicle that you want to clone. It could be a


player's vehicle or an AI vehicle.





Basically, this command creates an AI clone of the vehicle specified.


But this command can't just be used as it is. It would be much better


to set it equal to a variable first:





%clone = cloneVehicle(%vehicleId);





This way, we have defined %clone to be the orignal vehicle's clone. Now


we have a variable to work with. We can now use %clone anywhere throughout


the function to represent the clone. In the example below, we clone the


vehicle, then we teleport %clone to the desired coordinates (in this case,


nearby the %scanned).








function vehicle::onScan(%scanned, %scanner)


{


   %clone = cloneVehicle(%scanned);


   %x = getPosition(%scanned, x) + 100;


   %y = getPosition(%scanned, y) + 100;


   %z = getPosition(%scanned, z) + 100;


   setPosition(%clone, %x, %y, %z);


}








Now you should be able to clone vehicles. This command is pretty simple.











The randomTransport(); command:





randomTransport(%vehicleId, %x1, %y1, %x2, %y2);





Basically, this command is used to randomly teleport a vehicle within


a specified rectangular area. This rectangular area is specified by


2 sets of x and y coordinates. These sets of coordinates actually mark


the 2 opposite corners of this rectangle.








%vehicleId - This is the vehicle that you want to randomly teleport.





%x1 - This is the X coordinate of the FIRST corner of the rectangle. This


number must be HIGHER than the value of %x2.





%y1 - This is the Y coordinate of the FIRST corner of the rectangle. This


number must be HIGHER than the value of %y2.





%x2 - This is the X coordinate of the SECOND corner of the rectangle. This


number must be LOWER than the value of %x1.





%y2 - This is the Y coordinate of the SECOND corner of the rectangle. This


number must be LOWER than the value of %y1.








This command was also pretty simple. You may find it useful later on.











The setShapeVisibility(); command:





setShapeVisibility(%structure, true/false);





This command only works on structures. Basically, it is used to make


structures visible or invisible.





%structure - This is the structure that you want to make visible or


invisible.





true/false - In this spot, you put either true or false. True makes the


structure visible. False makes it invisible.








Example:





function structure::onAttacked(%attacked, %attacker)


{


   setShapeVisibility(%attacked, false);


}





In this function, whenever a structure is attacked, it becomes invisible.











The setVehicleSpecialIdentity(); command:





setVehicleSpecialIdentity(%vehicleId, on/off, %color);





This command is used to make vehicles glow a certain color, like in CTF


when a person has the flag.





%vehicleId - This is the vehicle that you want to make glow.





on/off - In this spot, you put either on or off. On makes the vehicle


glow. Off disables the glow.





%color - This represents the color that the vehicle will glow. It can be


set to yellow, blue, red, or purple. 








Example:





function vehicle::onAdd(%vehicleId)


{


   setVehicleSpecialIdentity(%vehicleId, on, blue);


}











The playSound(); command:





playSound(%player, "soundfile.wav", IDPRF_2D);





This command plays a sound to a particular player.





%player - This is the player ID of the person who hears the sound. If you


want to play the sound to everybody in the server instead of just one


person, just replace %player with 0.





"soundfile.wav" - This is the sound file that you want to play.





IDPRF_2D - This is the sound profile. It represents how the sound is to be


played. In this case, it plays a normal 2D sound effect. IDPRF_2D is the


only sound profile that I've been able to figure out how to use so far,


so for now, just use that.








In this example, a sound is played ONLY to a particular player:








function vehicle::onDestroyed(%destroyed, %destroyer)


{


   %player = playerManager::vehicleIdToPlayerNum(%destroyed);


   playSound(%player, "sfx_fog.wav", IDPRF_2D);


}








In this example, the same sound is played to all players in the server:








function vehicle::onDestroyed(%destroyed, %destroyer)


{


   playSound(0, "sfx_fog.wav", IDPRF_2D);


}











The damageObject(); command:





damageObject(%object, %damage);





This is one of the few commands that can damage/heal structures. It


can also damage vehicles.





%object - This is the structure or vehicle that you want to damage/heal.





%damage - This represents the amount of damage inflicted upon the structure


or vehicle. It is a numerical value. If the number is positive, it damages


objects. If the number is negative, it heals objects.








Example:





function structure::onScan(%scanned, %scanner)


{


   damageObject(%scanned, 100000);


}











The damageArea(); command:





damageArea(%object, %xOffset, %yOffset, %zOffset, %radius, %damage);





This command damages an area in a blast radius. This command is similar to


the blast() command, but unlike the blast() command, it can damage structures


as well as vehicles.





%object - This is the vehicle or structure that you want to damage.





%xOffset - This is the offset of the blast in the X axis. This is usually


set to 0 as default, and I recommend keeping it that way.





%yOffset - This is the offset of the blast in the Y axis. This is usually


set to 0 as default, and I recommend keeping it that way.





%zOffset - This is the offset of the blast in the Z axis. This is usually


set to 0 as default, and I recommend keeping it that way.





%radius - This represents the size of the blast radius in meters. This is


always a numerical value.





%damage - This represents the amount of damage that you want to inflict


upon the object. This is always a numerical value. I highly recommend only


using positive numbers. If you try to use negative numbers, you'll get some


strange side-effects.








Example:





function structure::onScan(%scanned, %scanner)


{


   damageArea(%scanned, 0, 0, 0, 150, 2500);


}





Most of the time, we keep the 3 offsets set to 0. These offsets aren't very


useful.











The schedule(); command:





schedule("Insert Command Here", %time);





This command is a very important one. Instead of just using a regular


command by itself, the schedule() command allows you to schedule certain


commands to be run at a later time.





"Insert Command Here" - This is where you put the command that you want to


schedule.





%time - This represents the amount of time (in seconds) that you want to


schedule the command for. It is always a numerical value.





Whatever command you put in the quotations will run at a later time.


Depending on long you set the time for, that will determine when the command


is run.








In this example, we schedule a global variable to be set equal to true 10


seconds after the function is activated:








function vehicle::onAttacked(%attacked, %attacker)


{


   schedule("$variable = true", 10);


}








If we wanted to schedule an actual command, we would have to do something


like this:








function vehicle::onAdd(%vehicleId)


{


   schedule("healObject(" @ %vehicleId @ ", -50000);", 20);


}








Because the command we used contains a local (percent) variable, we have to


separate the local variable from the rest of the command. If we do not do


this, we will get a syntax error in our schedule(). We use quotation


marks ("") to enclose the actual command. Then the variable is connected to


the rest of the command using @ signs. Try to think of it as separating the


%variable from the rest of the command. This method of linking the %variable


to the rest of the command with @ signs is called concatenation. It's really


not that complicated when you think about it. Be sure you learn how to do


this. We will be using concatenation much more later on. 





With global (dollar sign) variables, we don't need to concatenate. They


seem to do fine without concatenation, they won't give you syntax errors:








function vehicle::onAdd(%vehicleId)


{


   schedule("$global = false;", 7);


}





OR





function vehicle::onAdd(%vehicleId)


{


   schedule("healObject($global, -50000);", 7);


}








Concatenation is only necessary with local variables (variables that begin


with a percent sign). 








You may also get a syntax error when you try to use a command that uses


quotation marks, such as the say() command. The normal format of the say()


command looks like this:





say("Everybody", 1, "Hello everybody!");





But this command contains quotation marks, so we will have problems if we


try to use it with the schedule() command, because the schedule() command


also uses quotation marks. So we have to find a way to distinguish between


the say() command's quotation marks and the schedule() command's quotation


marks, so they won't conflict with each other. To fix this problem, we put


back-slashes before the say() command's quotation marks. Here's an example


of what I mean:





schedule("say(\"Everybody\", 1, \"Hello everybody!\");", 15);





Basically, the back-slashes designate that the quotation marks are to be


used as text characters instead of part of the actual schedule() code. If


you understand this now, you won't have any problems with it later on.











The setHudTimer(); command:





setHudTimer(%time, %direction, "Timer Title", %channel, %player);





This command is used to set a HUD timer on a player's screen.





%time - This represents the initial amount of time that you want to set on


the timer. It is a numerical value.





%direction - This represents the direction that the timer counts in. This


value can be set to either 1 or -1. If it is set to 1, then the timer will


count upwards. If it is set to -1, then the timer will count downwards.





"Timer Title" - This is the what you want the timer's title to be. The 


quotation marks around the text are required.





%channel - This represents the channel number that the timer will be set on.


This value can be set to either 1, 2, or 3. This is usually set to 1, but


this comes in handy when you want to set more than one timer at once. Each


timer would use a different channel. The maximum number of timers that can


be on your screen at any one time is 3. The channel number determines which


channel that your timer will use.





%player - This is the player whose screen that you want to set the HUD timer


on. If %player is replaced with a 0, then the timer will be set on everyone's


screen.





In this example, the 30 second HUD timer is set on a player's screen when


he spawns into the game:








function vehicle::onAdd(%vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   setHudTimer(30, -1, "HUD Timer 1" , 1, %player);


}








In the following example, we set a HUD timer on everyone's screen when a


certain structure is destroyed:








function generator::structure::onDestroyed(%destroyed, %destroyer)


{


   setHudTimer(60, -1, "Time To Escape:" , 1, 0);


}








We can also use the setHudTimer() command in conjunction with the


schedule() command to make something happen when the timer reaches zero.


In the following example, when a structure called "reactor" is destroyed,


a HUD timer is set on everyone's screen that tells them that they have


60 seconds to evacuate the area before the reactor explodes.








function reactor::structure::onDestroyed(%destroyed, %destroyer)


{


   setHudTimer(60, -1, "Time To Escape:" , 1, 0);


   schedule("damageArea(%destroyed, 0, 0, 0, 750, 1000000);", 60);


   schedule("playSound(0, \"sfx_fog.wav\", IDPRF_2D);", 60);


}








Notice how we use the schedule() command with setHudTimer() to schedule an


explosion after 60 seconds, which is also how long the HUD timer is set for.











The getVehicleNavMarker(); command:





getVehicleNavMarkerId(%vehicleId);





This command is used to get a vehicle's nav marker. Once you have the


vehicle's nav marker, you can set it equal to a variable and use it however


you want.





%vehicleId - This is the vehicle that you want to get the nav marker of.





The getVehicleNavMarkerId() command is used most often by setting it equal to


a variable first, as shown in the following example:








function navPorter::trigger::onEnter(%trigger, %vehicleId)


{


   %nav = getVehicleNavMarkerId(%vehicleId);


   %x = getPosition(%nav, x);


   %y = getPosition(%nav, y);


   %z = getPosition(%nav, z);


   setPosition(%vehicleId, %x, %y, %z);


}








In this function, when a vehicle enters the trigger, we get the vehicle's


nav marker and set it equal to %nav. Then we get the coordinates of the %nav


and then teleport the %vehicleId to those coordinates. Basically, whenever


a vehicle enters the trigger, the function teleports him to his nav point


(if he has one).





Here's another example of how this command can be used:








function navBoom::structure::onScan(%scanned, %scanner)


{


   %nav = getVehicleNavMarkerId(%scanner);


   blast(%nav, 150, 2500);


   say("Everybody", 1, getHudName(%scanner) @ " nuked his nav point!", "sfx_fog.wav");


}








In this function, when a person scans a certain structure, %nav is set equal


to the person's nav point. Then we blast() the navpoint (%nav), and say a


short message to everyone saying that that person nuked his nav point.











The reDrop(); command:





reDrop(%vehicleId);





This command is very simple. Basically, it re-drops the specified vehicle


at a random drop point. If the map is team play, then the vehicle will


re-drop at one of its team's drop points.





%vehicleId - This is the vehicle that you want to re-drop.





Example:








function vehicle::onAdd(%vehicleId)


{


   if(getTeam(%vehicleId) != *IDSTR_TEAM_BLUE)


   {


      setTeam(%vehicleId, *IDSTR_TEAM_BLUE);


      reDrop(%vehicleId);


   }


}








In this function, whenever a vehicle spawns into the game, we change its


team to blue, then re-drop the vehicle at one of its team's drop points.











The deleteObject(); command:





deleteObject(%object);





Basically, this command deletes the specified object. The object can be a


vehicle or a structure. 





%object - This is the vehicle or structure that you want to delete from the


map.











The playerManager::getPlayerCount() command:





playerManager::getPlayerCount();





This command gives you the current number of players in the server. It is


often used in if() checks and used by setting it equal to a variable.





Examples:








%count = playerManager::getPlayerCount();





In this example, a variable called %count is set equal to the current number of


players in the server.








function player::onAdd(%player)


{


   if(playerManager::getPlayerCount() > 3)


   {


      say("Everybody", 1, "There are now more than 3 players in the server.");


   }


}





In this example, we check to see if the current number of players in the


server is greater than 3.








These are both good examples of how to use this command.











The missionEndConditionMet(); command:





missionEndConditionMet();





This command makes the mission end. If your server is not dedicated, the


current map will restart. If your server is dedicated, the next map on the


server list will run.

















CHAPTER #7 - DEFINING OBJECTS WITH GLOBAL VARIABLES





Now we are going to learn how to set a structure or vehicle that has been


placed into the mission editor equal to a global variable, which can be


used later on. Take note, because this is one of the first steps to learning


how to use AI units (and to eventually make them respawn).





When you place an object such as a structure into the mission editor, you


are able to give it an object name. This object name can be used to define


what the object is in the script. We can then use a global variable and


set it equal to the structure in onMissionStart().








function onMissionStart()


{


   $structure1 = getObjectId("Missiongroup\\structure1");


}








In this function, we set a variable called $structure equal to some object


already placed in the map. We use the getObjectId() command to get the ID


number of the object (in this case the object is a structure). Inside the


getObjectId() command, we place the path that the object is found in between


a set of quotation marks. We use a set of double back-slashes between each


simgroup. The last part of the path is the object name of the structure. In


this case, the object name is structure1. If we had placed the structure


inside a simgroup folder in the mission editor, we must place the name of


the folder in the path so that Starsiege will be able to find the object:








function onMissionStart()


{


   $structure1 = getObjectId("Missiongroup\\Blue Base\\structure1");


}








In this case, if we had placed the structure in a folder called "Blue Base"


in the mission editor, we must put Blue Base in the path so that Starsiege


can find the object.





Once we define the structure as a global variable, we can then use


$structure1 anywhere in the script to represent that structure.








function structure::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $structure1)


   {


      say("Everybody", 1, "Structure #1 has been destroyed!");


   }


}





OR





function structure::onScan(%scanned, %scanner)


{   


   %player = playerManager::vehicleIdToPlayerNum(%scanner);





   if(%scanned == $structure1)


   {


      say(%player, 1, "This structure contains prototype cybrid weaponry.");


   }


}





OR





function promie::vehicle::onDestroyed(%destroyed, %destroyer)


{   


   damageObject($structure1, 100000);


}








All of these are good examples of how to use global variables that


represent certain objects already placed in the map.








AI vehicles can also be defined in the same way in onMissionStart().








function onMissionStart()


{


   $Exec1 = getObjectId("Missiongroup\\exec1");


}








Using the same method that we used to define the structures, we can define


AI vehicles as well.








function onMissionStart()


{


   $Exec1 = getObjectId("Missiongroup\\AI Vehicles\\exec1");


}








The same principles apply to defining vehicles that apply to defining


structures. If the vehicle was placed in a folder in the mission editor,


we must include the folder's name in the path. In this case, the folder's


name is "AI Vehicles", so we must include AI Vehicles in the path, so that


Starsiege can locate the AI vehicle.





Here are a few examples of how to use these variables in your script:








function onMissionStart()


{


   $Exec1 = getObjectId("Missiongroup\\exec1");


}





function vehicle::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $Exec1)


   {


      $Exec1 = cloneVehicle(%destroyed);


      setTeam($Exec1, getTeam(%destroyed));


      reDrop($Exec1);


      schedule("deleteObject(" @ %destroyed @ ");", 5);


   }


}








This short script contains a basic AI respawning function. When a vehicle


is destroyed, we check to see if it is the AI ($Exec1). If it is, then we


clone the vehicle and set the new clone equal to $Exec1. Then we set its


team to the team of the orignal AI. Then we re-drop the AI at a drop point.


Then the old AI unit is scheduled to be deleted 5 seconds later.





Here's another example:








function onMissionStart()


{


   $Exec1 = getObjectId("Missiongroup\\exec1");


}





function vehicle::onAttacked(%destroyed, %destroyer)


{


   if(%attacker == $Exec1)


   {


      healObject(%attacked, -50000);


   }


}








In this short script, the second function gives the AI vehicle one-hit kills.





These are only a few ways to use global variables to represent objects in


the map. In the next chapter, we will talk more about AI vehicles.

















CHAPTER #8 - AI VEHICLES





So far we've learned how to use AI vehicles to a certain degree, by defining


them as global variables in onMissionStart(). Now we will learn how to set


certain pilots to AI vehicles and how to give AI vehicles orders.





First of all, we will learn how to set pilots. When we want to make a custom


pilot for an AI vehicle, we use a pilot function. Pilot functions are very


unique. They are unlike regular functions in many ways. Here's an example


of a pilot function:








Pilot Iceman


{


   id = 28;


   name = "Major Iceman";


   skill = 0.9;


   accuracy = 0.9;


   aggressiveness = 1.0;


   activateDist = 850.0;


   deactivateBuff = 300.0;


   targetFreq = 2.0;


   trackFreq = 0.0;


   fireFreq = 0.2;


   LOSFreq = 0.2;


   orderFreq = 2.0;


};








If I were you, I would try to write you own pilot function. Just copy what


we have here and you can edit it how you want. 





All pilot functions begin with Pilot PilotName. The pilot name can be


whatever you want the pilot to be named. In this example, we named our pilot


Iceman.





The id is the ID number of the pilot. ID numbers start at 28 and go from


there. The ID number designates that pilot in Starsiege. Two pilots cannot


have the same ID number. If you want to have 2 different pilots in your


script, you must use different ID numbers (example: 28 and 29).





The name is that name of the pilot that shows up when you target the AI


vehicle. This can also be whatever you want. In this example, we named our


AI "Major Iceman".





The skill is how good you want the AI pilot to be. This can be set to a


number between 0.1 and 2.0, 0.1 being the easiest and 2.0 being the hardest.





The accuracy is how accurate the AI can fire its weapons. This can also be


set to a number between 0.1 and 2.0, 0.1 being the most innaccurate and


2.0 being the most accurate.





The aggressiveness is how aggressive the AI pilot will be. This can also be


set to a number between 0.1 and 2.0, 0.1 being the least agressive and


2.0 being the most agressive.





The activateDist is the distance at which the AI will sense an enemy and


begin to attack. Basically, it is radar range. This is set to a value that


is measured in meters.





The deactivateBuff is the distance beyond the activation distance that the


AI will stop attacking (or de-activate). . This is also set to a value that


is measured in meters.





The targetFreq is how often the AI checks to see if there is an enemy nearby.


This value is measured in seconds. It can be set from 0.1 (targeting every


0.1 seconds) to 2.0 (targeting every 2.0 seconds). There may or may not be a


limit on how high you can set this number.





The trackFreq is how often the AI asks himself: "If I leave my guns pointing


in this direction and at this angle, will I still hit my target?"


This can be set to a number between 0.0 and 2.0, the value being measured


in seconds. I highly recommend that this be kept at 0.0 though, or your


AI pilot will be very innaccurate.





The fireFreq is how often the AI fires his weapons. This value can be set


to a number between 0.1 and 2.0, the value being measured in seconds. The


lower the number, the more often the AI will fire his weapons per second.





The LOSFreq is how often the AI checks to see if there is an object


obstructing its line of sight between itself and its target. This value


can be set to a number between 0.1 and 2.0, the value being measured in


seconds. The lower the number, the more often the AI will check.





The orderFreq is how often the AI checks to see if it has been given an


order. This value can be set to a number between 0.1 and 2.0, the value


being measured in seconds.





Here's another example of a pilot:








Pilot Promie


{


   id = 42;


   name = "Prometheus";


   skill = 2.0;


   accuracy = 2.0;


   aggressiveness = 2.0;


   activateDist = 1200.0;


   deactivateBuff = 300.0;


   targetFreq = 0.1;


   trackFreq = 0.0;


   fireFreq = 0.1;


   LOSFreq = 0.1;


   orderFreq = 2.0;


};








In this example, we created an elite pilot......perfect for Prometheus.








Now that we've learned how to create a pilot, we will now learn how to set


these pilots to certain AI vehicles. We can do this by using the


setPilotId() command:





setPilotId(%vehicleId, %pilotId);





This command only works on AI vehicles.








%vehicleId - This is the vehicle that you want to set the pilot to.





%pilotId - This represents the ID number of the pilot that you want to use.








Note: You can set the same pilot to as many vehicles as you want.








Here's an example of how to use it:








Pilot Promie


{


   id = 42;


   name = "Prometheus";


   skill = 2.0;


   accuracy = 2.0;


   aggressiveness = 2.0;


   activateDist = 1200.0;


   deactivateBuff = 300.0;


   targetFreq = 0.1;


   trackFreq = 0.0;


   fireFreq = 0.1;


   LOSFreq = 0.1;


   orderFreq = 2.0;


};





function onMissionStart()


{


   $prom = getObjectId("Missiongroup\\herc1");


   setPilotId($prom, 42);


}








In this example, we made a pilot called promie (pilot #42). Then in


onMissionStart(), we defined a AI vehicle in the map to $prom. Then we set


pilot #42 to $prom.











Now we will learn how to use the order() command to order our AI vehicles


to do things. The order() command takes on many forms and does many different


things. Below is a list of the different types of order() commands and a


short description of each.








order(%vehicle1, attack, %vehicle2);





This command orders an AI vehicle to attack another vehicle.








order(%vehicleId, shutdown, true/false);





This command orders an AI vehicle to shut down or power up.








order(%vehicleId, cloak, true/false);





This command orders an AI vehicle to cloak or un-cloak. (AI vehicle must


have a cloak for this command to work)








order(%vehicleId, holdfire, true/false);





This command orders an AI vehicle to hold fire or to fire at will.








order(%vehicleId, holdposition, true/false);





This command orders an AI vehicle to hold its position or to not hold its


position.








order(%vehicle1, guard, %vehicle2);





This command orders an AI vehicle to guard another vehicle. An AI can also


be ordered to guard a structure.








order(%vehicleId, guard, %path);





This command orders an AI vehicle to guard a group of path markers. This


command basically makes AI's move.








order(%vehicleId, speed, high/low);





This command orders an AI vehicle to run or walk. This command is especially


useful when an AI is ordered to guard a path.








order(%vehicleId, clear);





This command clears an AI vehicle's current orders.











We won't cover all of these uses of the order() command, but we will


discuss how to use a few of the more important ones in more detail. Some


of the more important order() commands are the guard command and the attack


command. First, let's cover the guard command:





Now we will learn how to make AI vehicles follow a path. First of all, you


must create a simgroup folder in the mission editor. Then name the folder


whatever you want. Then inside the folder, place a few path markers (found


under the mission logic menu of the mission editor). These markers represent


the path which the AI vehicle will follow. If you only choose to put one


path marker, the AI will move to the path marker then stop. However, if you


in put more than one path marker, the AI vehicle will follow the markers in


order in a continuous loop. Once the Ai vehicle reaches the first marker in


the path, it will begin to move to the next marker, then the next one, etc.


When the AI vehicle reaches the last marker in the path, it will start over


and begin to move back towards the first marker in the list.





Once you've made a folder and a path for your AI vehicle to follow, you can


begin to script. First of all, you must define path in onMissionStart().


But how do we define an entire folder? So far, we've only learned how to


define individual objects, not an entire folder full of objects. To do this,


we must omit the getObjectId() command and use single front-slashes instead


of double back-slashes. The following 2 examples shows the differences in


how to define objects and how to define folders.








Use this format to define individual objects only:





function onMissionStart()


{


   $object = getObjectId("Missiongroup\\object");


}








Use this format to define entire folders:





function onMissionStart()


{


   $folder = "Missiongroup/folder";


}








We will use the second example to define our herc path because our herc path


consists of an entire folder full of path markers. If our herc path folder


is named path, we could do something like this:








function onMissionStart()


{


   $hercPath1 = "Missiongroup/path";


}








In this example, we defined $hercPath1 to be the herc path folder that we


already created in the map using the mission editor.





Now that we've defined what our path is, we can order our AI vehicle to


guard this path. When we do this, our AI vehicle will begin to follow the


path we created for him. Here's an example:








function onMissionStart()


{


   $herc1 = getObjectId("Missiongroup\\herc1");


   $hercPath1 = "Missiongroup/path";


   order($herc1, guard, $hercPath1);


}








Pretty simple, huh? There's not much more to it. This is how the order()


command is used to make a vehicle guard a path.





Now we can learn how to make an AI vehicle guard another vehicle. In the


following example, we order an AI vehicle to guard a person whenever that


person scans the AI.








function onMissionStart()


{


   $goad = getObjectId("Missiongroup\\goad");


}





function vehicle::onScan(%scanned, %scanner)


{


   if(%scanned == $goad)


   {


      order($goad, guard, %scanner);


   }


}








AI vehicles can also be ordered to guard structures or other AI vehicles


in much the same way. In the following example, we order one AI vehicle to


guard another AI vehicle:








function onMissionStart()


{


   $herc1 = getObjectId("Missiongroup\\herc1");


   $herc2 = getObjectId("Missiongroup\\herc2");


   order($herc1, guard, $herc2);


}








In the next example, we order an AI vehicle to guard a structure:








function onMissionStart()


{


   $herc = getObjectId("Missiongroup\\herc");


   $structure = getObjectId("Missiongroup\\structure");


   order($herc, guard, $structure);


}











Now let's learn how to use the attack command. This is basically used like


any other order() command. It's pretty simple to use, and can come in handy.


In the following example, we order an AI to attack a player's vehicle when


he spawns into the game:








function onMissionStart()


{


   $herc = getObjectId("Missiongroup\\herc");


}





function vehicle::onAdd(%vehicleId)


{


   order($herc, attack, %vehicleId);


}








The same thing can be done with 2 AI vehicles as well:





function onMissionStart()


{


   $herc1 = getObjectId("Missiongroup\\herc1");


   $herc2 = getObjectId("Missiongroup\\herc2");


   order($herc1, attack, $herc2);


}








You can also order an AI to attack a structure, but I think you get the idea.








Remember, you can stop any AI from carrying out its current orders by using


the clear command:





order(%vehicleId, clear);





Here's an example:








function onMissionStart()


{


   $herc = getObjectId("Missiongroup\\herc");


   $path = "Missiongroup/path";


   $currentOrder = "none";


   setTeam($herc, *IDSTR_TEAM_NEUTRAL);


}





function vehicle::onScan(%scanned, %scanner)


{


   if(%scanned == $herc)


   {


      if($currentOrder == "none")


      {


         $currentOrder = "guard";


         order($herc, guard, $path);


         say("Everybody", 1, "AI HERC: Roger that sir, following path.");


      }


      else if($currentOrder == "guard")


      {


         $currentOrder = "none";


         order($herc, clear);


         say("Everybody", 1, "AI HERC: Affirmative sir, orders canceled.");


      }


   }


}








In this short script, whenever the AI vehicle is scanned, he is ordered


to do something. If he currently has no orders, then he is ordered to guard


the path. If his current orders are to guard the path, then he will be


ordered to clear his orders (stop what he is doing).











Now what if we wanted to make our AI vehicle respawn and follow a path?


Well this should be easy, since by now we know how to do both. Here's an


example of how to do this:








function onMissionStart()


{


   $herc1 = getObjectId("Missiongroup\\herc1");


   $hercPath1 = "Missiongroup/path1";


   order($herc, guard, $path);


}





function vehicle::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $herc1)


   {


      $herc1 = cloneVehicle(%destroyed);


      setTeam($herc1, getTeam(%destroyed));


      reDrop($herc1);


      schedule("deleteObject(" @ %destroyed @ ");", 5);


   }


}








Now we will learn about a new function called vehicle::onArrived. Here is


an example of what the function looks like:








function herc1::vehicle::onArrived(%vehicleId, %where)


{


   if(%where == $pathmarker1)


   {


      INSERT COMMANDS HERE


   }


}








We use this function whenever we have an AI guarding something, whether it


be a path, a single path marker, a structure, or another vehicle. This


function activates whenever a certain AI vehicle reaches its destination


(what it's guarding). %vehicleId represents the AI vehicle, and %where


represents the object/pathmarker the AI is ordered to guard. We must include


an if() to check which path marker the AI has reached. In this case,


whenever the AI vehicle "herc1" arrives at $pathmarker1, the function is


activated.








And that's about it. Pretty easy once you learn the basics, eh?











On a side note, I think I should warn you about something. Whenever you


use a vehicle::onDestroyed function in your script, you may lose your


deathmessages in your map. Don't worry though, this can be fixed. Just


add the code from the vehicle::onDestroyed function found in DMstdLib.cs


to the vehicle::onDestroyed function in your map. The vehicle::onDestroyed


found in DMstdLib.cs looks like this:








function vehicle::onDestroyed(%destroyed, %destroyer)


{


   // left over from missionStdLib.cs


   vehicle::onDestroyedLog(%destroyed, %destroyer);


   


   // this is weird but %destroyer isn't necessarily a vehicle


   %message = getFancyDeathMessage(getHUDName(%destroyed), getHUDName(%destroyer));


   if(%message != "")


   {


      say( 0, 0, %message);


   }


   


   // enforce the rules


   if($server::TeamPlay == true)


   {


      if(


         (getTeam(%destroyed) == getTeam(%destroyer)) &&


         (%destroyed != %destroyer)


      )


      {


         antiTeamKill(%destroyer);


      }


   }   


}








You can merely copy the code inside this function an paste it inside your


vehicle::onDestroyed function below the code you've already written. So,


for example, if your vehicle::onDestroyed function looked like this:








function vehicle::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $herc1)


   {


      $herc1 = cloneVehicle(%destroyed);


      setTeam($herc1, getTeam(%destroyed));


      reDrop($herc1);


      schedule("deleteObject(" @ %destroyed @ ");", 5);


   }


}








All you would have to do is paste the code from the DMstdLib.cs 


vehicle::onDestroyed function inside your function below your code,


like this:








function vehicle::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $herc1)


   {


      $herc1 = cloneVehicle(%destroyed);


      setTeam($herc1, getTeam(%destroyed));


      reDrop($herc1);


      schedule("deleteObject(" @ %destroyed @ ");", 5);


   }





   // left over from missionStdLib.cs


   vehicle::onDestroyedLog(%destroyed, %destroyer);


   


   // this is weird but %destroyer isn't necessarily a vehicle


   %message = getFancyDeathMessage(getHUDName(%destroyed), getHUDName(%destroyer));


   if(%message != "")


   {


      say( 0, 0, %message);


   }


   


   // enforce the rules


   if($server::TeamPlay == true)


   {


      if(


         (getTeam(%destroyed) == getTeam(%destroyer)) &&


         (%destroyed != %destroyer)


      )


      {


         antiTeamKill(%destroyer);


      }


   }   


}








So your vehicle::onDestroyed function would look like the one above. That's


basically all you need to do to get your deathmessages to work. Again, this


is only necessary to do if you make your own vehicle::onDestroyed function 


in your map's script.

















CHAPTER #9 - MAKING YOUR OWN FUNCTIONS





So far, we've only used functions that were pre-defined by Starsiege. But


we can also make our own functions. You can name your function whatever you


want, and have it do anything you want it to do.





For example, what if we want to make a function called changeVariable()?








function changeVariable()


{





}








After we've made our function, we can make it do something.








function changeVariable()


{


   if($variable == false)


   {


      $variable = true;


   }


   else if($variable == true)


   {


      $variable = false;


   }


}








Now our function has a purpose. Whenever the function is run, it checks to


see what the variable is currently set to, then changes the variable to the


opposite of what it used to be set to. If the variable was set to false, it


changes to true. If the variable was set to true, it changes to false. The


name we picked for our function should describe what the function does.


(Although the name of the function doesn't really matter at all)





Now that we have our function, what do we do with it? Well, we can run our


function inside another function, using a command line. Here's an example


of what I mean:








function switch::structure::onScan(%scanned, %scanner)


{


   changeVariable();


}





function changeVariable()


{


   if($variable == false)


   {


      $variable = true;


   }


   else if($variable == true)


   {


      $variable = false;


   }


}








In the first function, we executed the second function (the one we made).


We used the name of the function followed by a semi-colon to convert it into


a command. Basically changeVariable(); is telling Starsiege to run that


function.








We can also use variables in our own custom-made functions. In the following


custom-made function punishVehicle(%attacker), we use %attacker because


that variable exists in the first function (temple::structure::onAttacked).








function temple::structure::onAttacked(%attacked, %attacker)


{


   punishVehicle(%attacker);


}





function punishVehicle(%attacker)


{


   %player = playerManager::vehicleIdToPlayerNum(%attacker);


   healObject(%attacker, -50000);


   messageBox(%player, "You have been punished for attacking the temple!");


   say("Everybody", 1, getHudName(%attacker) @ " has been punished!");


}








Our custom-made function is punishVehicle(%attacker). Its purpose is to


punish the vehicle who attacks the temple structure. Notice that we use


%attacker in our punishVehicle(%attacker) function because it was one of


the available variables in the temple::structure::onAttacked function.








You can also make your own functions that use more than one variable.


In the following example, we've made a function called 


cloneYourTarget(%scanned, %scanner).








function vehicle::onScan(%scanned, %scanner)


{


   %player = playerManager::vehicleIdToPlayerNum(%scanner);


   say(%player, 1, "Cloning your target...");


   cloneYourTarget(%scanned, %scanner);


}





function cloneYourTarget(%scanned, %scanner)


{


   %clone = cloneVehicle(%scanned);


   setTeam(%clone, getTeam(%scanner));


   reDrop(%clone);


   order(%clone, guard, %scanner);


}








In our cloneYourTarget(%scanned, %scanner) function, we used 2 variables,


%scanned and %scanner because both variables were available in the first


function (vehicle::onScan).








You can also make custom functions that use constant local variables.








function structure::onScan(%scanned, %scanner)


{


   %num = 2;


   cubeMyNumber(%num);


}





function cubeMyNumber(%num)


{


   $cubedNum = (%num * %num * %num);


   say("Everybody", 1, "The answer is" @ $cubedNum);


}








In this script, we made a custom function cubeMyNumber(%num). Its purpose


is to cube (raise to the 3rd power) whatever number we set %num to. In this


case, we set %num equal to 2. Then we run our function. The function takes


the value of %num and multiplies it by itself 3 times, and sets that value


equal to $cubedNum. Then the function tells you the answer in the chat


window using the say() command. In this case, since %num is equal to 2, the


answer would be 2 * 2 * 2 which equals 8. So our function would say


"The answer is 8". Pretty cool, huh?





Here's another example. In the following script, we've made a custom


function called respawnVehicle(%destroyed, %x, %y, %hercpath). The purpose


of our function is to respawn an AI vehicle when it is killed at specific


x and y coordinates, then order it to guard its herc path.








function onMissionStart()


{


   $herc1 = getObjectId("Missiongroup\\herc1");


   $herc2 = getObjectId("Missiongroup\\herc2");


   $path1 = "Missiongroup/path1";


   $path2 = "Missiongroup/path2";


}





function vehicle::onDestroyed(%destroyed, %destroyer)


{


   if(%destroyed == $herc1)


   {


      respawnVehicle(%destroyed, 243, 328, $path1);


   }


   else if(%destroyed == $herc2)


   {


      respawnVehicle(%destroyed, 657, -320, $path2);


   }


}





function respawnVehicle(%destroyed, %x, %y, %hercpath)


{


   %clone = cloneVehicle(%destroyed);


   setTeam(%clone, getTeam(%destroyed));


   setPosition(%clone, 0, 0, -1000);


   %z = getTerrainHeight(%x, %y);


   dropPod(%x, %y, (%z + 300), %x, %y, %z, %clone);


   order(%clone, guard, %hercpath);


   schedule("deleteObject(" @ %destroyed @ ");", 5);


}








In this example, we made our function do a number of useful things. But


those things couldn't be accomplished without certain variable values.


We recognize the fact that we need the values of the %destroyed, the


%x and %y coordinate numbers, and the correct %hercpath for the


corresponding AI vehicle that was destroyed. In the vehicle::onDestroyed


function, we check to see which AI vehicle was destroyed. Then we run


the command line respawnVehicle(%destroyed, %x, %y, %hercpath); with the


appropriate values for %x, %y, and %hercpath. If the %destroyed was $herc1,


then we used respawnVehicle(%destroyed, 243, 328, $path1); because we want


that AI to respawn at the coordinates 243, 328, and $herc1's herc path is


$path1. Then, when our custom function (respawnVehicle) is run, it takes


all of the values we inserted into the command line and uses them in the


place of the variables. In this case, our function clones the %destroyed


and sets it equal to %clone. Then it sets %clone's team to the team of the


%destroyed. Then it temporarily teleports it off the map. Next, our function


uses the x and y coordinates to determine where the herc should be dropped


in a drop pod. Then the AI is ordered to guard its appropriate hercpath.


And finally, the %destroyed (the old destroyed AI) is deleted 5 seconds


later. This example was probably one of the more useful ways to use


custom-made functions.








Custom functions can become a useful tool that will save you a lot of time,


especially when you have to repeat a series of commands several times in


your script. Instead of repeating these commands over and over every time


you need to do something, you can create a function that performs these


tasks every time you run the command line of the function. Trust me, custom


functions definitely save time. Basically, think of your custom function


as a command that you made. Your command runs the function you made, which


is basically a list of instructions that you customised.

















CHAPTER #10 - "DOT OPERATER" VARIABLES





So far we've only learned about global and local variables. But there is


also one more type of variable: dot-operator variables. Dot-operator


variables are variables that work for only one player or vehicle in the


game. Here are a few examples of dot-operator variables:





%vehicleId.variable





%player.variable





Basically, dot-operator variables are really 2 variables attached to each


other. The first part is the %player or %vehicleId (whoever owns the


variable). This is followed by a dot. The second part is the actual variable


that belongs to that player or vehicle. Think of it as SOMEONE's variable.


Somebody OWNS the variable. Therefore, in %player.variable, the %player


owns the variable. In %vehicleId.variable, the vehicle owns the variable.





For example, what if we wanted to keep a record of a person's HUD name


whenever their vehicle spawns into the game? We could use something like


this:








function vehicle::onAdd(%vehicleId)


{


   %vehicleId.name = getHudName(%vehicleId);


}








In this example, %vehicleId.name is set equal to the %vehicleId's HUD name.


Now the %vehicleId OWNS this variable. It represents his HUD name. We used


.name because it described what the variable stands for. You can use any


name for the .variable part though, it doesn't really matter.





Here's another example of a dot-operator variable:








function player::onAdd(%player)


{


   %player.kills = 0;


}





function vehicle::onDestroyed(%destroyed, %destroyer)


{


   %player = playerManager::vehicleIdToPlayerNum(%destroyer);


   %player.kills = %player.kills + 1;


}








In this example, we use a variable called %player.kills to keep track of


the %player's number of kills. When the %player first joins the server,


his kill number equals 0. Every time the %player destroys another vehicle,


%player.kills increases by 1. In this way, we can keep track the number


of kills the %player has.





NOTE: Dot-operator variables using %player exist for as long as the %player


remains in the server. However, dot-operator variables using %vehicleId


(or any vehicle ID for that matter) are deleted whenever the person's


vehicle is destroyed. This makes sense, because in the case of vehicle IDs,


the person's vehicle owns the variable. If it is destroyed, so is its


dot-operator variable.








Here's another example:








function player::onAdd(%player)


{


   %player.spawned = false;


}





function vehicle::onAdd(%vehicleId)


{


   %player = playerManager::vehicleIdToPlayerNum(%vehicleId);


   %player.spawned = true;


}





function vehicle::onDestroyed(%destroyed, %destroyer)


{


   %player = playerManager::vehicleIdToPlayerNum(%destroyed);


   %player.spawned = false;


}








In this example, we use %player.spawned to keep track of whether or not


a player is currently spawned. In player::onAdd, %player.spawned is set to


false, because obviously the player isn't spawned because he just did


join the server. Then in vehicle::onAdd, %player.spawned is set to true.


Then when the %player's vehicle is destroyed, %player.spawned is set back


to false.





Here's an example on how dot-operator variables can be used in a useful way:








function vehicle::onAdd(%vehicleId)


{


   %vehicleId.scanpower = "blast";


}





function toggleSwitch::structure::onScan(%scanned, %scanner)


{


   %player = playerManager::vehicleIdToPlayerNum(%scanner);


   if(%scanner.scanpower == "blast")


   {


      %scanner.scanpower = "heal";


      say(%player, 1, "Your power has been changed to healing.");


   }


   else if(%scanner.scanpower == "heal")


   {


      %scanner.scanpower = "blast";


      say(%player, 1, "Your power has been changed to blasting.");


   }


}





function vehicle::onScan(%scanned, %scanner)


{


   if(%scanner.scanpower == "blast")


   {


      blast(%scanned, 100, 10000);


      say("Everybody", 1, getHudName(%scanner) @ " blasted " @ getHudName(%scanned));


   }


   else if(%scanner.scanpower == "heal")


   {


      healObject(%scanned, 10000);


      say("Everybody", 1, getHudName(%scanner) @ " healed " @ getHudName(%scanned));


   }


}








In this script, whenever a person's vehicle spawns into the game, 


their %vehicleId.scanpower is set equal to "blast". However, the person


can change his scanning power by scanning the toggleSwitch structure. If


the person's %scanner.scanpower is set to "blast", it changes to "heal".


If his scanner.scanpower is set to "heal", then it changes to "blast".


Then, if that person scans another person's vehicle, then whatever power is 


currently selected will determine what happens to the scanned vehicle.


Depending on what a person's %scanner.scanpower equals, different things


will happen to the person he scanned. If it is set to "blast", then the


%scanned will be blasted, however if it is set to "heal", the %scanned will


be healed.





And that's basically all there is to dot-operator variables. Not too


complicated, is it? Just remember these 2 things:





1. If we have 2 dot-operator variables, one using a player ID and one using


a vehicle ID, they DO NOT mean the same thing. A variable called


%player.kills DOES NOT mean the the same thing as %vehicleId.kills, because


%player is a player ID and %vehicleId is a vehicle ID. In this first case,


the %player owns the variable, and in the second case, the %vehicleId owns


the variable.





2. If you have 2 dot-operator variables that both use vehicle ID's, then


they represent the same variable. Therefore %vehicleId.name means the same


thing as %scanner.name, which means the same thing as %scanned.name, which


means the same thing as %attacker.name, which means the same thing as


%attacked.name, which means the same thing as %destroyer.name, which means


the same thing as %destroyed.name, and so on. The reason this is true is


because all of these dot-operator variables use vehicle ID's. But remember,


as I said before, player ID's are not the same as vehicle ID's, so


%player.name DOES NOT mean the same thing as %vehicleId.name.

















CHAPTER #11 - CREATING NEW OBJECTS FROM THE SCRIPT





Now we will how to create new objects from the script using the newObject()


command. The newObject() command's format is as follows:





%object = newObject("Object Name", %objectType, %IDnumber);





This is the basic layout of the newObject() command. We will first learn how


to create new AI vehicles from the script.





%AIvehicle = newObject("BobtheAI", herc, 28);





In this example, we create a new AI vehicle set equal to %AIvehicle. We


named the AI vehicle's object name as "BobtheAI". We wanted the vehicle


to be a herc, so we specified herc as the object type. Then we used the


numer 28, because that is the ID number of an Executioner. ID numbers of


all of the vehicles in Starsiege can be found in defaultVehicles.cs in


your Starsiege\multiplayer folder.





Here's another example:








function structure::onScan(%scanned, %scanner)


{


   %tank1 = newObject("BattleTank", tank, 16);


   setPosition(%tank1, 0, 0, 200);


}








In this function, we created a tank named "BattleTank". In this particular


case, the ID number is 16, which means that this vehicle is a Knight's


Myrmidon.





We can also create flyers and drone vehicles using the newObject() command.


Here are a few examples:








function makeFlyer()


{


   $flyer1 = newObject("FireHawk", flyer, 9);


   setPosition($flyer, 0, 0, 200);


   order($flyer1, guard, $path1);


}





This function makes a flyer named "FireHawk". The ID number used is 9,


so this vehicle is a Banshee.








function createDrone()


{


   $drone = newObject("ConvoyLeader", drone, 65);


   setPosition($drone, 0, 0, 200);


   order($drone, guard, $path1);


}





This function makes a drone vehicle named "ConvoyLeader". The ID number


used is 65, so this vehicle is a cargo truck.








Once you've created your new AI vehicle, you can do whatever you want with


it. You can set its team, set its pilot, teleport it, order it to guard a


path, order it to attack something.....and the list goes on.





After an AI vehicle that you created using newObject() has died, it is a


good idea to delete the remains using the deleteObject() command. If you


don't do this, then you may end up with AI ghosts when the map restarts or


cycles. These AI ghosts are no fun. So if you are going to make AI vehicles


from the script using the newObject() command, then you need to delete


the AI vehicle whenever it is destroyed. If you don't want to do that, you


can always delete all your vehicles in onMissionEnd(). Here's what I mean:








function onMissionEnd()


{


   deleteObject($herc1);


   deleteObject($herc2);


   deleteObject($tank1);


   deleteObject($tank2);


   deleteObject($flyer1);


   deleteObject($drone1);


}








Make sure you get all of the AI you created. If you want to make the AI


vehicles respawn, you'll also have to deleteObject() the %destroyed AI


vehicle first. But we've already covered this in chapters 7 and 8.








You can also create new structures using the newObject() command. The format


for the newObject() command when creating structures is as follows:





%structure = newObject("Object Name", %objectType, %shapeFilename);





You can use registerObjects.cs in your Starsiege main directory to find the


object type and the shape filename. Look for the structure you want to


create in registerObjects.cs, then check to see whether it is considered


a StaticShape or a StaticInterior. Then, record the structure's shape


filename. If the structure is a StaticShape, then the filename should end


in the extension .DTS, however, if the structure is a StaticInterior, then


the filename should end in the extension .DIS.





Here are a few examples of how to create a structure:








function makeACrane()


{


   $structure1 = newObject("Crane", StaticShape, "hCrane.DTS");


   setPosition(%structure, 0, 0, getTerrainHeight(0, 0));


}





Notice that this structure is a StaticShape, therefore its shape filename


should end in .DTS, which it does.








function makeAGarage()


{


   $structure2 = newObject("TitanGarage", StaticInterior, "hTroophouseE.DIS");


   setPosition(%structure, 0, 0, getTerrainHeight(0, 0));


}





Notice that this structure is a StaticInterior, therefore its shape filename


should end in .DIS, which it does.








Then once you've created the new structure, you can do whatever you want


with it. You can set its team, animate it, teleport it, and more.





And just as I said before about AI vehicles, structures created from the


script using the newObject() command should also be deleted with the


deleteObject() command in onMissionEnd().








function onMissionEnd()


{


   deleteObject($structure1);


   deleteObject($structure2);


}








Now you should be able to create your own AI vehicles and structures from


the script.

















CHAPTER #12 - MAKING A GROUP OF COMMANDS RUN FOR EVERYBODY IN THE SERVER





We will now learn how to make a group of commands run for everybody in the


server. To do this, we must use a standard for() statement:








   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      INSERT COMMANDS HERE





   }








I highly recommend that you copy this basic example when you want to use it


in your script. Basically, this statement tells Starsiege to cycle through


every player in the server and execute the commands for each person. This


statement can be used in any function, even your own custom-made functions.


Here's an example:








function structure::onScan(%scanned, %scanner)


{


   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      INSERT COMMANDS HERE





   }


}








What if we wanted to kill everybody in the server when we scan a certain


structure? We could do something like this:








function killEverybody::structure::onScan(%scanned, %scanner)


{


   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      healObject(%vehicleId, -50000);





   }


}








What if we wanted to kill only people on yellow team? We could insert an


if() to check to see what team the person is on first.








function structure::onScan(%scanned, %scanner)


{


   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      if(getTeam(%vehicleId) == *IDSTR_TEAM_YELLOW)


      {


         healObject(%vehicleId, -50000);


      }


   }


}








If we wanted to use it in a custom-made function, we could do that too:








function killEverybodyInTheServer()


{


   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      if(getTeam(%vehicleId) == *IDSTR_TEAM_YELLOW)


      {


         healObject(%vehicleId, -50000);


      }


   }


}








We could also use it to set everybody's dot-operator variables to a certain


value:








function onMissionLoad()


{


   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      %player.points = 0;


      %vehicleId.spawned = false;


   }


}








In this example, each person in the server has their dot-operator variables


set to a certain value. In this case, %player.points was set equal to zero


and %vehicleId.spawned was set equal to false.








That is basically all there is to it. Like I said before, I highly recommend


that you copy and paste the for() statement into your scripts when you want


to use it. It's a pain to type, and it will save you much time and


frustration. The for() statement is the code in red text.








Here's another useful example:








function warnBlueTeam::trigger::onEnter(%trigger, %vehicleId)


{


   if(getTeam(%vehicleId) != *IDSTR_TEAM_BLUE)


   {


      $blueMessage = "TAC-COM: Warning: Enemy spotted on the base perimeter.";


      messageTeam(*IDSTR_TEAM_BLUE, $blueMessage);


   }


}





function messageTeam(%team, %message)


{


   %count = playerManager::getPlayerCount();


   for(%i = 0; %i < %count; %i++)


   {


      %player = playerManager::getPlayerNum(%i);


      %vehicleId = playerManager::playerNumToVehicleId(%player);





      if(getTeam(%player) == %team)


      {


         say(%player, 1, %message);


      }


   }


}








In this example script, we made a function called messageTeam(), which sends


a specific message to all players on the specified team, but only to that


team. In this case, whenever an enemy vehicle enters the "warnBlueTeam"


trigger, it activates the messageTeam() function which sends a warning


message to all players on blue team.





The for() statement can be used in many ways. If you learn to use it


effectively, it may help you to do what you want in your scripts later on.

















CHAPTER #13 - USING THE CONSOLE TO CHECK FOR SYNTAX ERRORS





In Starsiege, you can use the console to check for syntax errors in your


script. First active the console by clicking the white mark below the


decimal point in the SS version # on the main menu. Then go to the Create


Game screen. Select your map from the list. Then hit the tilde ~ key to


bring up the console. If you see a line that says SYNTAX ERROR LINE #,


then you have a syntax error in your script. Record the line # that the


syntax error was on. Then go into your script and scroll down to that


line. The error should be in this part of the script. Check to see if you


made a mistake in writing your code. A syntax error is nothing more than


an error caused by the way you wrote the script. It's probably a typo or a


careless mistake, such as leaving a semi-colon off the end of a command,


or forgetting to close a pair of parentheses, or leaving the closing


curly bracket off the end of a function. The syntax error could be caused


by a number of things. Once you think you have fixed the problem, go


back into the console to check the script for syntax errors again. If no


syntax error message appears, your script is good. However if the error


message appears again, you may have another error in your script. Check the


line # and go back to look at your script again. Repeat these steps until


your script contains no syntax errors.

















INDEX OF SCRIPTING SOURCES





The following is a list of scripting-related sources for you to use.


(listed in no particular order)








Orogogus's Scripting Website:  


http://home.san.rr.com/orogogus


This is the best scripting source I can recommend. It is a library of


scripting resources and can help you a lot.





Orogogus's Scripting Tutorials:


http://home.san.rr.com/orogogus/SSTutorial.html


These are a series of short scripting tutorials by Orogogus.





Orogogus's List of Functions:


http://home.san.rr.com/orogogus/functions.html


A fairly easy (and helpful) list of commands and functions.





Mike the Goad's List of Functions:


http://home.san.rr.com/orogogus/SScmd_lib.html


Another list of commands and functions (more advanced).





Starsiege Players Editing/Scripting Forum:


http://community.sierra.com/WebX?13@152.drmRaONZfXj^0@.ee6b597


This is a very good place to learn scripting. If you ever have a problem,


post a message here.





Men In Black Website:


http://www.starsiegemeninblack.cjb.net


The MIB website has several useful scripting-related downloads. I recommend


that you download the latest MIB Map Pack and take a look at some of the


map's scripts.





Men In Black Editing/Scripting Forum:


http://pub7.ezboard.com/fmibcommcentermibideacenter


This is the official MIB Editing/Scripting Forum. If you have a question,


post a message and I'll be glad to answer it.





Alpha Blue Modding/Scripting Guild:


http://www.planetstarsiege.com/alphablue


This is an organization of Starsiege modders and scripters. I recommend


joining if you really want to learn more about scripting.





Pincushion's Map Page: 


http://www.geocities.com/SoHo/Nook/9351/maps/index.html


Pincushion's maps and scripts. There are also a few other useful downloads


found here.





Pincushion's .cs Search Engine Index:


http://alphablue.sandwich.net/ssindex/index.html


This is a search engine where you can search for keywords in any .cs file


found in Starsiege. This can be very useful when looking for specific


commands.





Dolf Kooz's Maps and Scripts Page: 


http://www.nwlink.com/~jkudebeh/dolf


You may find these scripts to be useful references.





Dull's Harvester Page: 


http://www.geocities.com/imverydull/Harvester.html


You may also find it helpful to look at Harvester's script.


